

Mikro Styrepind



Bruger- vejledning

Mikro Styrepind

Brugervejledning

1. udgave

(C) 1985 by MIKRO VÆRKSTEDET

**Christian Wang
Niels Askær**

**Plottertegninger: Christian Wang
Illustration: Mona Askær
Fremstillet af: MIKRO VÆRKSTEDET
Odense Kommune**

**Printed in Denmark
ISBN 87-88832-05-8**

**MIKRO VÆRKSTEDET
Odense Kommune Mag. 4.afd.
Ryttervejen 2
5240 Odense NØ
Tlf. (09) 10 30 22**

Indhold

1. Introduktion
2. Funktionsprincip for MIKRO STYREPIND,
herunder tilslutning til en dataport
på en mikrodatamat
3. Tilslutningsdata for PICCOLINE,
herunder programeksempler i RcComal80
4. Tekniske specifikationer for
MIKRO STYREPIND
5. Undervisningsaktiviteter med
MIKRO STYREPIND

1 Introduktion

MIKRO STYREPIND er et "joystick" specielt udviklet til undervisningsbrug. **MIKRO STYREPIND** består af en XY-joystick, d.v.s. en "gearstang" med 4 kontakter, der afgiver et signal svarende til de 5 positioner: NEUTRAL, FREM, TILBAGE, HØJRE, VENSTRE.

MIKRO STYREPIND'en er desuden forsynet med 2 trykknapper hver med en signalledning. Den **grå** trykknop er **med lås** (artering), så knappen forbliver nede, d.v.s. aktiv, når den er trykket ned. Der skal således et yderligere tryk til for at få knappen op, d.v.s. passiv. Den **røde** trykknop er **uden lås** (ringetryk). Den skal holdes nede for at forblive aktiv, og slippes den, bliver den passiv.

I afsnit 5 beskrives nogle af de undervisningsmæssige muligheder styrepinden giver, her skal blot nævnes:

At give brugeren mulighed for direkte at styre ydre enheder (MIKRO TEGNEBIL, MIKRO ROBOTBIL og andet "robot"-udstyr).

At give brugeren mulighed for at generere programmer til styring af ydre enheder uden traditionel brug af tastatur.

At give brugeren mulighed for at generere tegninger på mikrodatamåtens skærm/printer/plotter.

At give elever uden de nødvendige motoriske færdigheder (bevægelseshandicappede) mulighed for at kommunikere v.h.a. mikrodatamåten.

At give elever uden de nødvendige sproglige forudsætninger mulighed for at kommunikere v.h.a. mikrodatamåten.

Styrepinden kan tilsluttes enhver mikrodatamat, hvor der er adgang til 6 parallelle bit samt stel. Styrepinden har indbyggede batterier, så det kan bruges uafhængigt af en evt. adgang til spændingsudgang fra mikrodatamaten.

2 Funktionsprincip


MIKRO STYREPIND'en består af en XY-Joystick med 4 omskiftere, samt 2 trykknapper hver med en omskifter.

Disse omskiftere sender enten et signal bestående af en lav elektrisk spænding (0 volt) eller et signal bestående af en høj elektrisk spænding (ca. 5 volt), til den mikrodatamat hvor styrepinden er tilsluttet.

For at gøre styrepinden uafhængig af strømforsyning fra mikrodatamaten er der monteret et udskifteligt batteri, der sørger for den nødvendige elektriske spænding til de "høje" signaler. Dette batteri skal ved normal brug udskiftes ca. 1 gang om året.

Hver af de 6 omskiftere virker efter dette princip:

Fig. 1



PASSIV = HØJ

AKTIV = LAV

Selve styrepinden kan stå i 5 positioner, som vist på fig.2-6:

NEUTRAL




Fig. 2

FREM




Fig. 3

TILBAGE




Fig. 4

HØJRE




Fig. 5

VENSTRE




Fig. 6

Styrepinden tilsluttet en mikrodatamat

Når styrepinden tilsluttes en mikrodatamat, skal de 6 omskiftere forbindes til hver sin bit i en parallelport. Der skal altså mindst 6 parallelle bits og en stelforbindelse til, for at styrepinden og de 2 trykknapper kan udnyttes.

Når styrepinden tilsluttes en "dataport", vil de høje signaler medføre, at de enkelte bits får værdien 1, og de lave signaler vil sætte bit-værdierne til 0. Hvis man fra et program aflæser bitværdien 1, betyder det, at den pågældende omskifter ikke er aktiveret - den er passiv. Hvis bitværdien 0 aflæses fra et program, betyder det, at den pågældende omskifter er aktiv.




Fig. 7

Mikro Styrepind tilsluttet en dataport hvor høj=1 og lav=0. På fig. 7 er styrepinden i positionen "FREM", den grå trykknop er nede, altså aktiv, og den røde trykknop oppe, altså passiv. I ovenstående tilfælde vil følgende decimale tal kunne aflæses fra programmet:

Bit 0:	0×2^0	= 0
Bit 1:	1×2^1	= 2
Bit 2:	1×2^2	= 4
Bit 3:	1×2^3	= 8
Bit 4:	0×2^4	= 0
Bit 5:	1×2^5	= 32

Decimale tal: **46**

Den følgende tabel er en oversigt over de decimale måletal, der kan aflæses fra et program, der benytter **MIKRO STYREPIND**'en tilsluttet en dataport som vist på fig.7.

MIKRO STYREPIND tilsluttet:

ET PARALLEL IN/OUT-STIK
PÅ ENHVER MIKRODATAMAT
PORT: HØJ=1 OG LAV=0

RØD KNAP UDEN LÅS	GRÅ KNAP MED LÅS	STYREPIND	KNAPPER RØD GRÅ		STYREPIND				MÅLE- TAL DECIMALT
			BIT 5	BIT 4	BIT 3	BIT 2	BIT 1	BIT 0	
OPPE	OPPE	NEUTRAL	1	1	1	1	1	1	63
OPPE	OPPE	FREM	1	1	1	1	1	0	62
OPPE	OPPE	TILBAGE	1	1	1	1	0	1	61
OPPE	OPPE	HØJRE	1	1	1	0	1	1	59
OPPE	OPPE	VENSTRE	1	1	0	1	1	1	55
OPPE	NEDE	NEUTRAL	1	0	1	1	1	1	47
OPPE	NEDE	FREM	1	0	1	1	1	0	46
OPPE	NEDE	TILBAGE	1	0	1	1	0	1	45
OPPE	NEDE	HØJRE	1	0	1	0	1	1	43
OPPE	NEDE	VENSTRE	1	0	0	1	1	1	39
NEDE	OPPE	NEUTRAL	0	1	1	1	1	1	31
NEDE	OPPE	FREM	0	1	1	1	1	0	30
NEDE	OPPE	TILBAGE	0	1	1	1	0	1	29
NEDE	OPPE	HØJRE	0	1	1	0	1	1	27
NEDE	OPPE	VENSTRE	0	1	0	1	1	1	23
NEDE	NEDE	NEUTRAL	0	0	1	1	1	1	15
NEDE	NEDE	FREM	0	0	1	1	1	0	14
NEDE	NEDE	TILBAGE	0	0	1	1	0	1	13
NEDE	NEDE	HØJRE	0	0	1	0	1	1	11
NEDE	NEDE	VENSTRE	0	0	0	1	1	1	7

(32) (16) (8) (4) (2) (1)

Fig. 8

3 MIKRO STYREPIND tilsluttet Piccoline

Styrepinden kan umiddelbart tilsluttes "local printer" stikket på Piccoline CPU-enheden. Når nærværende brugervejledning indeholder et specielt afsnit om tilslutning til Piccoline, er det fordi, vi anbefaler, at styrepinden *ikke* tilsluttes dataporten i "local printer" stikket, men kontrolporten. Dette giver nemlig mulighed for at benytte dataporten til f.eks. MIKRO TEGNEBIL og MIKRO ROBOTBIL eller andet "robot" udstyr, der styres fra dataporten.

Forskellen mellem "local printer" stikkets dataport og kontrolport er, bortset fra portnummeret:

- at dataporten registrerer et udefra kommende højt signal som 1, og et lavt signal som 0 (som vist på fig.7), og
- at kontrolporten registrerer et udefra kommende højt signal som 0, og et lavt signal som 1 (kontrolporten er **inverteret** i forhold til dataporten).

Det betyder, at **MIKRO STYREPIND**'en tilsluttet kontrolporten fungerer som vist på fig.9:




Fig. 9

Mikro Styrepind tilsluttet Piccoline's kontrolport i "local printer" stikket, hvor høj=0 og lav=1. På fig. 9 er styrepinden i positionen "FREM", den grå trykknop er nede, altså aktiv, og den røde trykknop oppe, altså passiv. I ovenstående tilfælde vil følgende decimale tal kunne aflæses fra programmet:

Bit 0:	1×2^0	= 1
Bit 1:	0×2^1	= 0
Bit 2:	0×2^2	= 0
Bit 3:	0×2^3	= 0
Bit 4:	1×2^4	= 16
Bit 5:	0×2^5	= 0

Decimale tal: 17

Når styrepinden er tilsluttet kontrolporten, vil en passiv omskifter altså svare til en bitværdi på 0 og en aktiv omskifter svare til en bitværdi på 1.

Den følgende tabel er en oversigt over de decimale måletal, der kan aflæses fra et program, der benytter **MIKRO STYREPIND**'en tilsluttet Piccolinens kontrolport (port nr. 608) i "local printer" stikket, som vist på fig.9.

MIKRO STYREPIND tilsluttet:

Mikro: PICCOLINE
Stik : LOCAL PRINTER
Port : 608 (INVERTERET)

			KNAPPER		STYREPIND				MÅLE- TAL DECIMALT
RØD KNAF UDEN LÅS	GRÅ KNAF MED LÅS	STYREPIND	BIT 5	BIT 4	BIT 3	BIT 2	BIT 1	BIT 0	
OPPE	OPPE	NEUTRAL	0	0	0	0	0	0	0
OPPE	OPPE	FREM	0	0	0	0	0	1	1
OPPE	OPPE	TILBAGE	0	0	0	0	1	0	2
OPPE	OPPE	HØJRE	0	0	0	1	0	0	4
OPPE	OPPE	VENSTRE	0	0	1	0	0	0	8
OPPE	NEDE	NEUTRAL	0	1	0	0	0	0	16
OPPE	NEDE	FREM	0	1	0	0	0	1	17
OPPE	NEDE	TILBAGE	0	1	0	0	1	0	18
OPPE	NEDE	HØJRE	0	1	0	1	0	0	20
OPPE	NEDE	VENSTRE	0	1	1	0	0	0	24
NEDE	OPPE	NEUTRAL	1	0	0	0	0	0	32
NEDE	OPPE	FREM	1	0	0	0	0	1	33
NEDE	OPPE	TILBAGE	1	0	0	0	1	0	34
NEDE	OPPE	HØJRE	1	0	0	1	0	0	36
NEDE	OPPE	VENSTRE	1	0	1	0	0	0	40
NEDE	NEDE	NEUTRAL	1	1	0	0	0	0	48
NEDE	NEDE	FREM	1	1	0	0	0	1	49
NEDE	NEDE	TILBAGE	1	1	0	0	1	0	50
NEDE	NEDE	HØJRE	1	1	0	1	0	0	52
NEDE	NEDE	VENSTRE	1	1	1	0	0	0	56

(32) (16) (8) (4) (2) (1)

Fig. 10

Styring af tegnebil og robotbil ved hjælp af styrepinden

Hvis styrepinden ønskes tilsluttet samtidig med MIKRO TEGNEBIL, MIKRO ROBOTBIL eller andet eksternt udstyr, skal dataporten (hvor tegnebil og robotbil tilsluttes), og kontrolporten (hvor styrepinden tilsluttes) adskilles til hver sit stik.

Dette gøres med et ekstrakabel, som vist på fig. 11 (se i øvrigt afsnit 4):




Fig. 11

Med ekstrakablet er det muligt at styre tegnebil eller robotbil ved hjælp af styrepinden via et program, der evt. kan udarbejdes så styringsprocessen "huskes" af mikrodatamaten med henblik på en eller flere senere gentagelser eller lagring på diskette.

Programeksempler i RcComal80

Programmet "STPIND01" skriver på skærmen måletallet fra styrepinden:

```
0010 // "STPIND01"
0020 CLOSE
0030 // datastrøm fra styrepind //
0040 OPEN FILE 2, "/608/port", READ
0050 REPEAT
0060     // aflæsning af signal fra styrepind //
0070     måletal:= ORD(GET$(2,1))
0080     // udskrivning af måletal fra styrepind //
0090     PRINT måletal
0100 UNTIL FALSE
```

Programmet "STPIND02" gør det muligt at styre skrivepletten (en stjerne) rundt på skærmen med styrepinden. Når den grå trykknop er trykket ned, tegnes en streg efter skrivepletten. Når den grå trykknop er oppe slettes evt. streger på skærmen. Når den røde knap trykkes ned, flyttes skrivepletten tilbage til midten af skærmen.

```
0010 // "STPIND02"
0020
0030 opstart
0040 tegn_m_styrepind
0050 END
0060
0070 PROC tegn_m_styrepind
0080     PRINT AT(x,y); " ";
0090     REPEAT
0100         glx:= x; gly:= y
0110         måletal:= ORD(GET$(2,1))
0120         CASE måletal OF
0130             WHEN 1 // frem uden streg
0140                 IF y>=2 THEN y:= y-1
0150                 glplet$:= " "
0160             WHEN 2 // tilbage uden streg
0170                 IF y<=22 THEN y:= y+1
0180                 glplet$:= " "
0190             WHEN 4 // højre uden streg
0200                 IF x<=78 THEN x:= x+1
0210                 glplet$:= " "
```

```
0220      WHEN 8 // venstre uden streg
0230          IF x>=2 THEN x:= x-1
0240          glplet$:= " "
0250      WHEN 17 // frem med streg
0260          IF y>=2 THEN y:= y-1
0270          glplet$:= " "
0280      WHEN 18 // tilbage med streg
0290          IF y<=22 THEN y:= y+1
0300          glplet$:= " "
0310      WHEN 20 // højre med streg
0320          IF x<=78 THEN x:= x+1
0330          glplet$:= " "
0340      WHEN 24 // venstre med streg
0350          IF x>=2 THEN x:= x-1
0360          glplet$:= " "
0370      OTHERWISE
0380          IF måletal>=32 THEN x:= 40; y:= 12
0390      ENDCASE
0400      IF glx<>x OR gly<>y THEN
0410          PRINT AT(glx, gly);glplet$;
0420      ENDIF
0430      PRINT AT(x,y);"*";
0440      FOR pausegennemløb:= 1 TO 100 DO
0450          NEXT pausegennemløb
0460      UNTIL FALSE
0470  ENDPROC tegn_m_styrepind
0480
0490  PROC opstart
0500      ENABLE escape
0510      PRINT CHR$(12);
0520      CLOSE
0530      // datastrøm fra styrepind //
0540      OPEN FILE 2, "/608/port", READ
0550      x:= 40; y:= 12
0560      DIM glplet$ OF 1
0570      glplet$:= " "
0580  ENDPROC opstart
0590
0600  PROC escape HANDLER
0610      CLOSE FILE 2
0620      PRINT CHR$(12)
0630      END
0640  ENDPROC escape
```

Programmet "STPIND03" gør det muligt at styre **MIKRO TEGNEBIL**'en direkte med styrepinden. Hver gang programmet indlæser et måletal fra styrepinden sendes 40 kørepulser (linie 340-370), svarende til 10 graders drejning af tegnebilen, eller til ca. 14 mm frem- eller tilbagekørsel. Når den grå knap er nede er tegnepennen på tegnebilen også nede, og tegnebilen vil tegne. Når den grå knap er oppe, er tegnepennen også oppe, og tegnebilen tegner således ikke. Den røde knap udnyttes ikke fra dette program.

```
0010 // "STPIND03"
0020
0030 opstart
0040 tegn_m_tegnebil
0050 END
0060
0070 PROC tegn_m_tegnebil
0080     REPEAT
0090         // signal fra styrepind indlæses //
0100         måletal:= ORD(GET$(2,1))
0110         CASE måletal OF
0120             WHEN 0 // pennen oppe
0130                 høj_pulstal:= 0; lav_pulstal:= 0
0140             WHEN 1 // frem med pennen oppe
0150                 høj_pulstal:= 41; lav_pulstal:= 32
0160             WHEN 2 // tilbage med pennen oppe
0170                 høj_pulstal:= 13; lav_pulstal:= 4
0180             WHEN 4 // højre med pennen oppe
0190                 høj_pulstal:= 45; lav_pulstal:= 36
0200             WHEN 8 // venstre med pennen oppe
0210                 høj_pulstal:= 9; lav_pulstal:= 0
0220             WHEN 16 // pennen nede
0230                 høj_pulstal:= 64; lav_pulstal:= 64
0240             WHEN 17 // frem med pennen nede
0250                 høj_pulstal:= 105; lav_pulstal:= 96
0260             WHEN 18 // tilbage med pennen nede
0270                 høj_pulstal:= 77; lav_pulstal:= 68
0280             WHEN 20 // højre med pennen nede
0290                 høj_pulstal:= 109; lav_pulstal:= 100
0300             WHEN 24 // venstre med pennen nede
0310                 høj_pulstal:= 73; lav_pulstal:= 64
0320             OTHERWISE
0330             ENDCASE
```

```
0340     IF måletal<=24 THEN
0350         FOR pulsnr:= 1 TO 40 DO
0360             PRINT FILE 1: CHR$(høj_pulstal);
0370             PRINT FILE 1: CHR$(lav_pulstal);
0380         NEXT pulsnr
0390     ENDIF
0400     UNTIL FALSE
0410 ENDPROC tegn_m_tegnebil
0420
0430 PROC opstart
0440     ENABLE escape
0450     PRINT CHR$(12);
0460     PRINT AT(18,10);"STYR MED STYREPINDEN !!"
0470     ENABLE escape
0480     CLOSE
0490     // datastrøm til tegnebil //
0500     OPEN FILE 1,"/592/port", WRITE
0510     // datastrøm fra styrepind //
0520     OPEN FILE 2,"/608/port", READ
0530 ENDPROC opstart
0540
0550 PROC escape HANDLER
0560     CLOSE
0570     PRINT CHR$(12)
0580     END
0590 ENDPROC escape
```

4 Tekniske specifikationer

Kabinet

Materiale: PVC

Længde: 150 mm

Bredde: 80 mm

Højde uden styrepind: 80 mm

Højde incl. styrepind: 170 mm

Styrepind

Mærke: Honneywell

Type: M2551NM

Styrepinden er af industristandard, og den er således meget modstandsdygtig overfor mekaniske og kemiske påvirkninger.

Alle 6 udgangssignaler til mikrodatamaten er med "elektrisk pull up", forsynet fra 3 stk. batterier af typen: 1.5 V transistorbatteri international R6 Hellesen type 716. Batterierne bør skiftes ca. 1 gang årligt, eller når lysdioderne ved styrepinden ikke lyser eller lyser meget svagt.

Kabel

Længde: ca. 2.5 meter

Stikdefinition til Piccoline:

(Stiktype: 25 ben "Canon")

PORT (BITNR.)	BEN NR.:	STYREPIND FUNKTION:
608(0)inv	11	FREM
608(1)inv	10	TILBAGE
608(2)inv	15	HØJRE
608(3)inv	12	VENSTRE
608(4)inv	13	GRÅ TRYKKNAP MED LÅS
608(5)inv	1	RØD TRYKKNAP UDEN LÅS
0 VOLT	18	STELFORBINDELSE

Ekstra tilbehør

Adapterkabel så både MIKRO STYREPIND og MIKRO TEGNEBIL/MIKRO ROBOTBIL/MIKRO FABRIK kan tilsluttes samtidig til PICCOLI-NE'ns "local printer" stik.

Stikdefinition for adapterkabel (alle 3 stik: 25 ben "Canon"):

PORT (BITNR.):	HANSTIK LOCAL PRINTER:	HUNSTIK MIKRO STYREPIND	HUNSTIK MIKRO TEGNEBIL/ROBOTBIL:
592(0)	2	2	-
592(1)	3	3	-
592(2)	4	4	-
592(3)	5	5	-
592(4)	6	6	-
592(5)	7	7	-
592(6)	8	8	-
592(7)	9	9	-
608(0)inv	11	-	11
608(1)inv	10	-	10
608(2)inv	15	-	15
608(3)inv	12	-	12
608(4)inv	13	-	13
608(5)inv	1	-	1
608(6)inv	16	-	16
608(7)inv	17	-	17
STEL	18-25	18	18

5 Undervisningsaktiviteter med MIKRO STYREPIND

I dette afsnit er medtaget forslag og ideer til anvendelsesområder for MIKRO STYREPIND'en. Afsnittet er ikke en gennemgang af alle de muligheder, anvendelsen af styrepinden giver, men et igangsættende idegrundlag for brugen af den.

Som det fremgår af afsnit 3, er brugen af styrepinden til styring af udstyr, der kan tilkobles mikrodatamaten en oplagt mulighed. Den enkleste måde at gøre det på er ved at styre bilen/robotten uden mulighed for at gemme eller repetere det forløb, der netop er blevet udført. Programmet i afsnit 3 "STPIND03" kan anvendes til det.

Industrirobotter programmeres ofte ved, at operatøren styrer robotten manuelt for "at vise" hvordan den skal bevæge sig. Disse bevægelser omdanner robotens styresystem til et program, der herefter styrer og kontrollerer robotens bevægelser.

Et sådant system kan efterlignes af et program med følgende egenskaber:

Programmet til programmering af en ydre enhed ved hjælp af styrepinden genererer programlinierne på skærmen samtidig med at styrepinden samt den ydre enhed aktiveres.

Eksempelvis:

Føres styrepinden frem, skrives der på skærmen **FREM ()**. I parentes angiver en tæller, hvor langt den ydre enhed skal bevæge sig frem. Denne tæller er aktiveret, så længe styrepinden er i positionen **FREM**. Samtidig med at tælleren er aktiveret, kører bilen/robotten frem, indtil styrepinden føres tilbage i midterpositionen. Er tegnebilen kørt for langt frem, føres styrepinden **TILBAGE**. Tallet i parentes reduceres så tilsvarende. Bliver tallet negativ ændres kommandoen til **TILBAGE**. Når den ønskede status for den ydre enhed er opnået, trykkes der på den røde knap for at fortælle mikrodatamaten, nu er den programlinie ok. På tilsvarende vis fungerer program og styrepind for de andre retninger. Tegneredskaber/gribere m.v. aktiveres ved hjælp af den grå trykknapp, samtidig skrives der på skærmen **PEN NED**.

Nedenstående er et eksempel på et program genereret ved hjælp af styrepinden og programmet "MVSTYR":

```
1 FREM (100)
2 PEN NED
3 TILBAGE (50)
4 VENSTRE (45)
5 FREM (25)
6 HØJRE (120)
7 FREM (10)
8 PEN OP
```

Dette program kan afprøves, og det kan gemmes på diskette m.v.. Med dette program eller et tilsvarende vil man kunne efterligne programmeringen af de såkaldte "playback-robotter".

MIKRO STYREPIND'en samt et program som ovenfor beskrevet vil desuden sætte børn uden de nødvendige sproglige egenskaber istand til at "skrive" EDB-programmer, der kan styre forskelligt udstyr, der er tilkoblet microdatamaten.

Ønsker du ikke selv at lave et program som her beskrevet, kan det købes hos MIKRO VÆRKSTEDET.

MIKRO STYREPIND'en kan anvendes sammen med traditionelle undervisningsprogrammer. Undervisningsprogrammer skal ikke være dårlige efterligninger af "pløk-ned-spil", men det er et faktum, at spil, der bruger joystick, især har børns store interesse. Denne interesse vil med fordel, kunne udnyttes som en motiverende faktor i forbindelse med forskellige typer undervisningsprogrammer. Det vil f.eks. være oplagt at erstatte brugen af cursor pilene med brugen af styrepinden.

Mikrodatamater bør kunne anvendes og betjenes af alle. Mange mennesker med et motorisk handicap er desværre forhindret i at kunne bruge datamaten, fordi al kommunikation mellem brugeren og datamaten som regel skal foregå via et tastatur. Hvis et EDB-program udformes derefter, vil styrepinden kunne erstatte et tastatur, og den handicappede, hvad enten det drejer sig om et barn eller en voksen, vil kunne anvende mikrodatamaten.

